



BIOMETRICS INFORMATION

(You're 95% likely to need this information)

PAMPHLET NO. # 3

DATE: February 24, 1988

SUBJECT: ANOVA using SAS: Specifying error terms

Analysis of variance problems of any complexity rarely use the default error for all tests. When using SAS to analyse such problems, it is necessary to supply TEST statements specifying the appropriate error term for all sources not to be tested with the sampling error term.

Consider the following repeated measures design:

<u>Source of Variation</u>		<u>Degrees of Freedom</u>	<u>Error Term</u>
Ecosystem	E	2	P(ET)
Treatment	T	1	P(ET)
ExT		2	P(ET)
Plots	P(ET)	6	S(PET)
Seedlings	S(PET)	228	--
Date	D	4	DxP(ET)
DxE		8	DxP(ET)
DxT		4	DxP(ET)
DxExT		8	DxP(ET)
DxP(ET)		24	DxS(PET)
DxS(PET)		912	--

Four different error terms are required for this analysis. One of these terms, DxS(PET), is the sampling error, and will be the default error term used automatically by SAS to test all sources in the entire model. Since DxS(PET) is not the appropriate error term for all sources, it is necessary to specify the correct error term for all sources other than those to be tested by DxS(PET). This is accomplished via the TEST statement. The required coding for the SAS analysis of variance procedure is as follows:

```

PROC ANOVA;
  CLASS E T P S D;
  MODEL HEIGHT = E T E*T P(E*T) S(P*E*T) D D*E D*T D*E*T D*P(E*T);
  TEST H=E T E*T          E=P(E*T);
  TEST H=P(E*T)          E=S(P*E*T);
  TEST H=D D*E D*T D*E*T  E=D*P(E*T);
  MEANS S(P*E*T) D*P(E*T) / DUNCAN;
  MEANS E T E*T          / DUNCAN E=P(E*T);
  MEANS P(E*T)          / DUNCAN E=S(P*E*T);
  MEANS D D*E D*T D*E*T  / DUNCAN E=D*P(E*T);
  TITLE 'REPEATED MEASURES ANALYSIS OF VARIANCE';

```

Each TEST statement specifies one or more sources to be tested by a specified error term. The sources to be tested are listed following H= (H stands for hypothesis), while the specified error term follows E= (E stands for error).

Note that there are 4 MEANS statements. This is because the appropriate error term must also be specified for DUNCAN multiple range tests. The first MEANS statement does not specify an error term, thus the default is used. The remaining MEANS statements specify the correct error terms for the sources they apply to.

One other point should be noted: the default error term, D_xS(PET) is not specified in the MODEL statement. SAS will calculate this value automatically and simply call it ERROR. Including it in the MODEL statement will greatly increase the amount of memory required by the program, and may exceed the capabilities of the computer.

CONTACT: John Thornton
387-3021